The following is the analytical solution for the diffusion in composite planar geometry: (For more details refer to "Lithium Intercalation in Core-Shell Materials–Theoretical Analysis")

```
> restart:with(plots):Digits:=15:with(RootFinding):with(IntegrationTools):
>
>
>
>
> #User defined parameters:
```

Description of parameters: beta^2=D2/D1, alpha=R1/R2, kappa=c1*/c2*,delta=current density

```
> constants:=[beta=2,alpha=0.8,gamma=1000,kappa=1.5,delta=-.25];
```

$$constants := [\beta = 2, \alpha = 0.8, \gamma = 1000, \kappa = 1.5, \delta = -0.25]$$

Number of eigenvalues to be used (Higher the better!!!)

```
> NN:=40:
>
>
>
> #solution
> param1:=[theta[n]=alpha*lambda[n]*beta,phi[n]=lambda[n]*(alpha-1)];
```

$$param1 := [\theta_n = \alpha \lambda_n \beta, \phi_n = \lambda_n (\alpha - 1)]$$

Eigenvalue Equation

```
> eigeneqn:=-kappa/tan(phi[n])+beta/tan(theta[n])+lambda[n]/gamma;
```

$$eigeneqn := -\frac{\kappa}{\tan(\phi_n)} + \frac{\beta}{\tan(\theta_n)} + \frac{\lambda_n}{\gamma}$$

coverting eigevale equation in to sin and cos form in order to get numerically stable eigenvalues

```
> subs(param1,simplify(convert(eigeneqn,sincos))*(sin(phi[n])*sin(theta[n])*gamma)=0);
```

$$\beta \cos(\alpha \lambda_n \beta) \sin(\lambda_n (\alpha - 1)) \gamma - \kappa \cos(\lambda_n (\alpha - 1)) \sin(\alpha \lambda_n \beta) \gamma + \lambda_n \sin(\lambda_n (\alpha - 1)) \sin(\alpha \lambda_n \beta) = 0$$

Function to return Eigenequation (necessary for using Nextzero function of Maple)

```
> KKK:=lambda->subs(lambda[n]=lambda,beta*cos(alpha*lambda[n]*beta)*sin(lambda[n]*(alpha-1))*gamma-kappa*cos(lambda[n]*(alpha-1))*sin(alpha*lambda[n]*beta)*gamma+lambda[n]*sin(lambda[n]*(alpha-1))*sin(alpha*lambda[n]*beta));
```

$$KKK := \lambda \to \mathrm{subs}(\lambda_n = \lambda, \beta \cos(\alpha \lambda_n \beta) \sin(\lambda_n (\alpha - 1)) \gamma - \kappa \cos(\lambda_n (\alpha - 1)) \sin(\alpha \lambda_n \beta) \gamma + \lambda_n \sin(\lambda_n (\alpha - 1)) \sin(\alpha \lambda_n \beta))$$

```
> ChiParam:=
  [
  chi[1]=kappa*gamma*cos(phi[n])-lambda[n]*sin(phi[n]),
```

```
chi[2]=cos(theta[n])*sin(theta[n])+theta[n],
chi[3]=cos(phi[n])*sin(phi[n])+phi[n],
chi[4]=beta*gamma*cos(theta[n])+lambda[n]*sin(theta[n])
];
```

$$ChiParam := [\chi_1 = \kappa\,\gamma\cos(\phi_n) - \lambda_n \sin(\phi_n),\ \chi_2 = \cos(\theta_n)\sin(\theta_n) + \theta_n,\ \chi_3 = \cos(\phi_n)\sin(\phi_n) + \phi_n,\ \chi_4 = \beta\,\gamma\cos(\theta_n) + \lambda_n \sin(\theta_n)]$$

```
> Aneqn:=A[n]=2*beta*kappa*delta*cos(theta[n])*cos(phi[n])/(chi[1]*l
ambda[n]*(chi[2]*beta*kappa*cos(phi[n])^2-chi[3]*chi[4]^2/gamma^2)
);
```

$$Aneqn := A_n = \frac{2\,\beta\,\kappa\,\delta\cos(\theta_n)\cos(\phi_n)}{\chi_1 \lambda_n \left(\chi_2\,\beta\,\kappa\cos(\phi_n)^2 - \frac{\chi_3\,\chi_4^2}{\gamma^2}\right)}$$

```
> OtherParam:=[
k[1]=-kappa*delta/(alpha*(kappa-1)+1),
k[2]=-delta/(alpha*(kappa-1)+1),

a[1]=
delta*kappa/6/(alpha*(kappa-1)+1)^2*
((kappa-1)*(alpha^3*beta^2+3*alpha^3-6*alpha^2+3*alpha)+(1-beta^2)
*2*alpha^3-3*alpha^2*(1-beta^2)+1)
+(alpha-1)*alpha*kappa*delta/(alpha*(kappa-1)+1)^2/gamma,

a[2]=
delta/6/(alpha*(kappa-1)+1)^2
*(6*alpha^3*(kappa-1)^2-(2*alpha^3*beta^2-6*alpha^3-3*alpha)*(kapp
a-1)+2*alpha^3*(1-beta^2)+1)
+kappa*alpha^2*delta/gamma/(alpha*(kappa-1)+1)^2];
```

$$OtherParam := \left[k_1 = -\frac{\kappa\,\delta}{\alpha\,(\kappa-1)+1},\ k_2 = -\frac{\delta}{\alpha\,(\kappa-1)+1},\ a_1 = \frac{\delta\,\kappa\,((\kappa-1)(\alpha^3\beta^2 + 3\,\alpha^3 - 6\,\alpha^2 + 3\,\alpha) + 2\,(-\beta^2+1)\,\alpha^3 - 3\,\alpha^2(-\beta^2+1) + 1)}{6\,(\alpha\,(\kappa-1)+1)^2} + \frac{(\alpha-1)\,\alpha\,\kappa\,\delta}{(\alpha\,(\kappa-1)+1)^2\,\gamma},\right.$$
$$\left. a_2 = \frac{\delta\,(6\,\alpha^3(\kappa-1)^2 - (2\,\alpha^3\beta^2 - 6\,\alpha^3 - 3\,\alpha)(\kappa-1) + 2\,(-\beta^2+1)\,\alpha^3 + 1)}{6\,(\alpha\,(\kappa-1)+1)^2} + \frac{\kappa\,\alpha^2\,\delta}{\gamma\,(\alpha\,(\kappa-1)+1)^2}\right]$$

```
> ##################
```

Final Solution

Part1 of solution (without the infinite sum)

```
> x1part1:=1/2*beta^2*k[1]*X^2+k[1]*tau+a[1];
```

$$x1part1 := \frac{1}{2}\beta^2 k_1 X^2 + k_1 \tau + a_1$$

```
> x1part2:=A[n]*gamma*beta*kappa*cos(phi[n])*cos(lambda[n]*beta*X)*e
  xp(-lambda[n]^2*tau);
```

$$x1part2 := A_n \gamma \beta \kappa \cos(\phi_n) \cos(\lambda_n \beta X) \mathbf{e}^{\left(-\lambda_n^{\ 2} \tau\right)}$$

```
>
```

Part2 # Infinite sum part

```
> x2part1:=1/2*k[2]*X^2-(delta+k[2])*X+k[2]*tau+a[2];
```

$$x2part1 := \frac{1}{2} k_2 X^2 - (\delta + k_2) X + k_2 \tau + a_2$$

```
> x2part2:=A[n]*(beta*gamma*cos(theta[n])+lambda[n]*sin(theta[n]))*c
  os(lambda[n]*(X-1))*exp(-lambda[n]^2*tau);
```

$$x2part2 := A_n (\beta \gamma \cos(\theta_n) + \lambda_n \sin(\theta_n)) \cos(\lambda_n (X - 1)) \mathbf{e}^{\left(-\lambda_n^{\ 2} \tau\right)}$$

```
> ########################
>
>
> #Calculation of x1 and x2 expressions using numerical eigenvalues
>
>
```

First Eigenvalue

```
> Lam[1]:=NextZero(subs(constants,lambda[n]=lambda,eval(KKK)),0);
```

$$Lam_1 := 1.68972332983742$$

```
> for i from 2 to NN do
  Lam[i]:=NextZero(subs(constants,eval(KKK)),Lam[i-1]);od:
```

List of all the eigenvalues

```
> ListLambda:=[seq(lambda[i]=Lam[i],i=1..NN)]:
>
> subs(constants,OtherParam);
```

$[k_1 = 0.267857142857143, k_2 = 0.178571428571429, a_1 = -0.151755102040816,$
$a_2 = 0.0129727891156463]$

```
> # Computation of Expressions for x1 and x2
> expr11:=0:expr22:=0:
  for i from 1 to NN do

  NumChiparam:=evalf(subs(param1,n=i,ListLambda,constants,ChiParam))
```

```
;
NumAn:=evalf(subs(param1,n=i,ListLambda,constants,NumChiparam,Aeq
n));

expr11:=expr11
+subs(param1,n=i,NumChiparam,ListLambda,NumAn,OtherParam,constants
, x1part2):

expr22:=expr22
+subs(param1,n=i,NumChiparam,ListLambda,NumAn,OtherParam,constants
, x2part2):
od:
> expr1:=expr11+subs(OtherParam,constants,x1part1):
> expr2:=expr22+subs(OtherParam,constants,x2part1):
>
>
>
> #plots:
> p3:=plot(evalf(subs(tau=0,expr1)),X=0..eval(alpha,constants)):p4:=
plot(evalf(subs(tau=0,expr2)),X=eval(alpha,constants)..1):display(
p3,p4);

>
> q1:=plot3d(expr1,X=0..subs(constants,alpha),tau=1/4000..0.1,axes=b
oxed):
> q2:=plot3d(expr2,X=subs(constants,alpha)..1,tau=1/4000..0.1,axes=b
oxed):
> with(plots):
> display(q1,q2);
```




```
> #backcheck (Mass conservation)
```

Integral of concentration over X

```
> IntC:=int(expr1,X=0..eval(alpha,constants))+int(expr2,X=eval(alpha
  ,constants)..1):
```

Integration of Flux up to that point

```
> IntF:=-subs(constants,delta*tau);
```

$$IntF := 0.25\ \tau$$

```
> pp1:=plot(IntC,tau=0..1,style=point,color=black,symbolsize=20):
> pp2:=plot(IntF,tau=0..1,thickness=3,color=red):
```

Plot of both integration (Both plots should be identical, if not increase the number of eigenvalues to be used)

```
> display(pp1,pp2);
```




```
>
>
>
```